

Función CONCATENATEX

Compatibilidad

Microsoft Excel
Excel ≥ 2016

★★★★★

Power BI Desktop
PBI ≥ Nov 2016

★★★★★

DIRECTQUERY: C.Calculadas  Medidas 
ROW LEVEL SECURITY: 

Int. Contexto

Contexto de Filtro
Tiene en cuenta el contexto de filtro

★★★★★

Contexto de Fila
Ignora contextos de filas previos
Tiene en cuenta su contexto de fila

Categorías

Según Proceso Interno
Iteración

★★★★★

Según Resultado
Escalar

Recursos de Aprendizaje



MAGÍSTER EN LEGUAJE DAX

100% en Vivo - Más Información:


→ [Capacitación OnLine] ←

<https://bit.ly/3bzlkG0>

Descripción

La función **CONCATENATEX** concatena o combina en una única cadena de caracteres todos los valores parciales que «salieron» como resultado de una expresión que se evalúa fila a fila en una tabla, separados por el delimitador señalado en su tercer parámetro.

Sintaxis



1 Tabla

Puede ser algo tan sencillo como el nombre de una tabla, ejemplo: **Pedidos** o algo más elaborado mediante una expresión de tipo tabla, ejemplo: **VALUES('Pedidos'[Tipo de Compra])**. Aquí es donde se ejecutará el recorrido fila a fila.

 Iterador

 Expresión Tipo Tabla

Tipo

Obligatorio

Atributo

No Repetible 

Videos Sobre ITERADORES



En el seminario grabado en directo se exponen los fundamentos de las funciones de iteración escalares, es decir, **SUMX**, **AVERAGEX**, **MEDIANX**, etc.

Con lo previo, en el seminario se crean medidas o cálculo de agregación con restricciones tipo O, tipo Y y combinaciones de los dos, donde, aprovechamos para ver las diversas formas de aplicar un O, con la función **OR**, el operador **|**, el operador **IN**, como implementar una operación tipo NOTIN, tanto con constructor de tabla y lista externa.

[→ Ver Video Aquí ←](#)

Capacitación Gratuita de DAX de 16 horas:

LEGAJE DAX DE CERO A GUERRERO

[VER AQUÍ](#)

2 Expresión

Una expresión escalar de tipo texto, aritmético, algebraico o booleana para ser evaluada en el recorrido fila a fila.

Contexto de Fila

Tipo

Obligatorio

Atributo

No Repetible

3 Delimitador

El carácter o texto delimitador utilizado para la concatenación. Si el parámetro es omitido no utiliza delimitador, uniendo valor tras valor sin espacios.

Tipo

Opcional

Atributo

No Repetible

4 Expresión Ordenar Por

Una expresión que será utilizada para ordenar la tabla.

Contexto de Fila

Tipo

Opcional

Atributo

Repetible

5 Orden

El orden para ser aplicado, indicado así: **0** o **FALSE** o **DESC** para orden descendente (*mayor a menor*), **1** o **TRUE** o **ASC** para orden ascendente (*menor a mayor*).

Si el parámetro es omitido toma **1** como valor por defecto **1**, es decir, ordenación ascendente (**ASC**).

Tipo

Opcional

Atributo

Repetible

Valor Que Retorna

Un valor único escalar de tipo **STRING** que representa el texto con los valores concatenados.

Ejemplos

- Ejemplo 1 – Parámetro de Tabla Sencillo:

Para el modelo de DISPRODUCTOS LD encontramos una matriz con los países en el área de filas, y un año específico seleccionado en una segmentación de datos, concretamente 1999.

- ☛ Crear una medida para ser arrastrada a la matriz que muestre los distintos productos vendidos (SKUs) en los diversos países separados por coma (,).

Una alternativa de solución se consigue así:

```

1. PrdctVnds = -- Expresión para una medida
2. CONCATENATEX (
3.     VALUES ( Pedidos[SKU] ) ,
4.     Pedidos[SKU],
5.     ""
6. ) -- Explorar archivo de Power BI asociado para ver la solución.
    
```

- Ejemplo 2 – Parámetro de Tabla con Más Elementos:

El primer parámetro (*Tabla*) como se ha corroborado en el ejemplo anterior, admite funciones de tabla, y puede ser un expresión sencilla o tan compleja como sea necesaria.

- ☛ Por ejemplo, supongamos que deseamos la misma medida, pero siempre remover de la lista los productos CB01 y L07.

Una alternativa para solucionarlo se presenta a continuación:

```

1. PrdctVnds-2 = -- Expresión para una medida
2. CONCATENATEX (
3.     EXCEPT ( VALUES ( Pedidos[SKU] ) , { "CB01", "L07" } ) ,
4.     Pedidos[SKU],
5.     ""
6. ) -- Explorar archivo de Power BI asociado para ver la solución.
    
```



“ La presente ficha técnica cuenta con un archivo de Power BI asociado con el modelo de Disproductos, donde podrás ver cada medida y su expresión DAX. Descargar CartasDAX.Com ”

↓ Descargar Aquí



www.CartasDAX.Com

Los ejemplos 3, 4 y 5 utilizan fuertemente diversas funciones de texto del lenguaje DAX, cada una cuenta con su ficha técnica, además, la web www.CartasDAX.Com te brinda la opción de explorar las funciones por categorías.

Por lo que, si deseas explorar todas las funciones de texto, encontraras una página dedicada:

→ Explorar Funciones de Textot Aquí ←

Otros Recursos de Aprendizaje



Experto en Lenguaje DAX

Formación Completa

→ [Visitar Libro] ←



Int. Excel y Power BI

Capítulo número 17

→ [Visitar Libro] ←



Excel Free Blog

Canal de YouTube

→ [Visitar Canal] ←

• Ejemplo 3 — Expresión con Más Elementos:

El segundo parámetro (*Expresión*) no tiene que ser necesariamente el llamado a una columna existente en la tabla pasada en el primer parámetro, puede ser una expresión escalar con más elementos.

☛ *Ejemplo: Crear una medida que será arrastrada a la matriz, y que muestre los distintos productos vendidos (SKUs) en los diversos países separados por coma (,), pero, todas las letras en minúsculas y sustituyendo el 0 por vacío.*

Una alternativa de solución se consigue con la ayuda de [SUBSTITUTE](#) y [LOWER](#), de la siguiente manera:

```

1. PrdctVnds-3 =
2. -- Expresión para una medida
3. CONCATENATEX (
4.     VALUES ( Pedidos[SKU] ) ,
5.     SUBSTITUTE ( LOWER ( Pedidos[SKU] ) , "0" , BLANK ( ) ) ,
6.     ","
7. ) -- Explorar archivo de Power BI asociado para ver la solución.
    
```

• Ejemplo 4 — Delimitador de Salto de Línea:

El tercer parámetro (*Delimitador*) es comúnmente empleado para tener una interrupción visual de la concatenación, empero, puede ser cualquier carácter unicode válido (véase la ficha técnica de [UNICHAR](#)).

☛ *Por ejemplo: Si queremos mostrar la lista de SKUs en líneas diferentes, se puede resolver así:*

```

1. PrdctVnds-4 =
2. -- Expresión para una medida
3. CONCATENATEX (
4.     VALUES ( Pedidos[SKU] ) ,
5.     Pedidos[SKU] ,
6.     UNICHAR ( 10 )
7. ) -- Explorar archivo de Power BI asociado para ver la solución.
    
```

Seminario Sobre Variables en DAX



La solución al ejemplo 5 puede parecer ser de alta complejidad, no obstante, es sencilla, pues es extensa en número de líneas, pero simple en su lógica de construcción.

La solución del ejemplo 5 utiliza variables para dividir en pasos el problema, con el apoyo de las funciones de texto del lenguaje DAX.

El estudio de variables es un aspecto fundamental, por lo que, si no tienes conocimiento de ellas o las manipulas a un nivel básico, es recomendable que dediques un espacio para estudiar la grabación 100% gratuita del seminario: [Guía Definitiva a Variables en Lenguaje DAX](#).

[→ Ver Video Aquí ←](#)

• Ejemplo 5 – Ordenación Descendente:

La ordenación viene de la mano de los parámetros 4 y 5. Por defecto la ordenación ocurre ascendentemente (ASC), en otras palabras, de menor a mayor o de A a Z. Sin embargo, si queremos desplegar el resultado de forma descendente, es decir, de mayor a menor o Z a A, debemos indicar explícitamente los parámetros 4 (*Expresión Ordenar Por*) y 5 (*Orden*).

☞ Por ejemplo: Crear una medida que será arrastrada a la matriz que muestre los distintos productos vendidos (SKUs) en los diversos países separados por el icono: "🐉", pero, la primera letra del SKU en mayúsculas y el resto en minúsculas, además, sustituir el 0 por nada. Presentar el resultado en orden descendente.

Una alternativa de solución utilizando diversas funciones de texto, puede ser así:

```

1. PrdctVnds-5 = -- Expresión para una medida
2. CONCATENATEX (
3.     VALUES ( Pedidos[SKU] ) ,
4.     VAR SKUActual = Pedidos[SKU]
5.     VAR PriemraLetra =
6.         LEFT ( SKUActual )
7.     VAR RestoDeLetras =
8.         MID ( SKUActual , 2 , LEN ( SKUActual ) )
9.     VAR RestoDeLetrasMinuscultas =
10.        LOWER ( RestoDeLetras )
11.     VAR ReemplazarCerosPorNada =
12.        SUBSTITUTE ( RestoDeLetrasMinuscultas , "0" , BLANK ( ) )
13.     VAR SKUFinal =
14.        CONCATENATE ( PriemraLetra , ReemplazarCerosPorNada )
15.     RETURN
16.        SKUFinal ,
17.        "🐉" ,
18.        Pedidos[SKU] , DESC
19. ) -- Explorar archivo de Power BI asociado para ver la solución.
    
```

MAGÍSTER EN LENGUAJE DAX — NIVEL 2

La capacitación **Magíster en Lenguaje DAX** del equipo de **Power Skill Tech** cuenta con tres niveles, cada uno de 30 horas.

El nivel más popular es el 1, no obstante, si cuentas con un conocimiento amplio en los fundamentos, como: **contexto de fila**, **contexto de filtro**, **funciones de iteración**, **CALCULATE avanzado**, **transición de contextos** y **manejo de tablas de calendario con funciones de inteligencia de tiempo**, entonces, el nivel 2 es para ti, ya que parte de dicho conocimiento y explora otras temáticas y aplicaciones de carácter avanzado.

[\[Más Información Aquí\]](#)

• Ejemplo 6 — Ordenación Personalizada:

La ordenación personalizada con la función **CONCATENATEX** es perfectamente posible, ahora bien, requerimos de una columna en la tabla del primer parámetro que señale mediante un número entero la posición del elemento según esa ordenación personalizada.

☛ *Por ejemplo: En la tabla Pedidos del modelo de Disproductos, tenemos una columna denominada Orden que indica esa disposición personalizada de los SKUs. Se demanda una medida que arroje la lista de SKUs vendidos (sin ningún cambio), pero presentándola con el orden personalizado.*

La solución con **CONCATENATEX** y con apoyo de la función **SUMMARIZE**, es como sigue:

```
1. PrdctVnds-6 = -- Expresión para una medida
2. CONCATENATEX (
3.     SUMMARIZE ( Pedidos , Pedidos[SKU] , Pedidos[Orden] ) ,
4.     Pedidos[SKU] ,
5.     " " ,
6.     Pedidos[Orden], ASC
7. ) -- Explorar archivo de Power BI asociado para ver la solución.
```

ORDEN EN VARIOS NIVELES

Si se requiere ordenación con más niveles, la función **CONCATENATEX** lo maneja a la perfección, dado que después del quinto parámetro los argumentos se repiten por parejas: **<Expresión Ordenar Por> + <Orden>** de forma indefinida. Tan sólo recuerda incluir esas columnas adicionales de ordenación en el primer parámetro (*Tabla*) de la función **CONCATENATEX**.

Las aplicaciones con la función **CONCATENATEX** en apoyo con otras funciones del lenguaje DAX son diversas y a todas luces interminables. Veamos un par de ejemplos.



Degustación DAX

- Ejemplo 7 – Top 3 Países con Mayores Ingresos y sus Porcentajes:

Escenario: En una matriz tenemos Años y Mes Nombre de la tabla de Calendario en filas, y la medida Ingresos Tot en el área de valores como punto de referencia.

- ☛ Se requiere una medida que devuelva los 3 países que generan los mayores ingresos, pero además que muestre el porcentaje de participación al lado del país, la medida será arrastrada a la matriz anterior.

La solución:

```

1. Top3PaisesConIngresos =
2. -- Expresión para una medida
3. VAR IngresosTotales =
4.     [Ingresos Tot]
5. VAR IngresosPorPais =
6.     ADDCOLUMNS ( VALUES ( Pedidos[País] ) , "@Ingresos" , [Ingresos Tot] )
7. VAR ParticipacionPorPais =
8.     ADDCOLUMNS ( IngresosPorPais , "@Pct" , DIVIDE ( [ @Ingresos] , IngresosTotales ) )
9. VAR PaisyParticipacion =
10.    ADDCOLUMNS (
11.        ParticipacionPorPais ,
12.        "@Resultado" ,
13.        Pedidos[País]
14.        & "( "
15.        & FORMAT ( [ @Pct] , "0.0%" )
16.        & ")"
17.    )
18. VAR Top3Paises =
19.     TOPN ( 3 , PaisyParticipacion , [ @Ingresos] )
20. VAR Resultado =
21.     CONCATENATEX ( Top3Paises , [ @Resultado] , "," & UNICHAR ( 10 ) )
22. RETURN
23.     Resultado
24. -- Explorar archivo de Power BI asociado para ver la solución.
25. -- NOTA:
26. -- El resaltado en verde de las columnas temporales (las del @), es una ayuda visual sólo de la ficha técnica.
    
```

Nuevamente es una expresión larga más no compleja.

- Ejemplo 8 – Mostrar Filtros Aplicados a un Reporte:

En muchos reportes se dejan ciertos filtros por fuera de la vista inmediata del lector, por ejemplo, en el panel de filtros.

Un caso típico viene de la mano con segmentación de datos sincronizada en una página diferente, lo cual puede ser confuso, dado que no tiene presente los elementos seleccionados.

[\[Puedes Estudiar Cómo Filtrar Múltiples Páginas con Una Única Segmentación de Datos: Aquí\]](#)

☛ *En el modelo de datos de Disproductos crear una matriz con el campo País en el área de filas y la medida Ingresos Tot en el área de valores, en otra página crear una segmentación de datos con el campo SKU de la tabla SKUProductos con los siguientes elementos seleccionados: B01, B02, CB01, CC01, L01, L02, L03 y L04.*

*El objetivo: Crear una medida que liste los elementos seleccionados de la segmentación de datos en una tarjeta o cuadro de texto, sin embargo, si todos los elementos (SKUs) están seleccionados, entonces, que muestre el texto: **“Todos los SKUs seleccionados”**, dado que no sería una experiencia óptima mostrar una lista enorme de SKUs.*

Algo similar va a suceder si hay muchos SKUs seleccionados, si bien no todos, una cantidad mayor a 5 elementos puede causar una experiencia subóptima de lectura, por ello, si hay más de 5 elementos seleccionados, entonces, que liste los 5 primeros SKUs y además que indique el número restante de SKUs que no se listan, e invitar a revisar la página con la segmentación para más detalles.

La solución como es de esperar no sólo requiere de [CONCATENATEX](#), sino también de funciones como: [FILTERS](#), [COUNTRROWS](#), [ALL](#), [TOPN](#) e [IF](#), lo que va a producir una expresión DAX bastante extensa, con varios pasos intermedios con la ayuda de variables, pero nuevamente, la expresión no es compleja es tan sólo extensa en número de líneas.


```

1.  SKUSeleccionados =
2.  -- Expresión para una medida
3.  VAR NumeroDeElementosMaximosAMostrar =
4.      5
5.  VAR SKUs =
6.      FILTERS ( SKUProductos[SKU] )
7.  VAR NumeroDeSKUsSeleccionados =
8.      COUNTROWS ( SKUs )
9.  VAR NumeroDeSKUsTotales =
10.     COUNTROWS ( ALL ( SKUProductos[SKU] ) )
11.  VAR EstanTodosLosSKUsSeleccionados = NumeroDeSKUsSeleccionados = NumeroDeSKUsTotales
12.  VAR Top5SKUs =
13.     TOPN ( NumeroDeElementosMaximosAMostrar, SKUs, SKUProductos[SKU], ASC )
14.  VAR ListaDeSKUsSeleccionados =
15.     CONCATENATEX (
16.         Top5SKUs,
17.         SKUProductos[SKU],
18.         ", "
19.     )
20.  VAR Resultado =
21.     "Filtrado por: "
22.     & IF (
23.         EstanTodosLosSKUsSeleccionados,
24.         "Todos los SKU Seleccionados",
25.         IF (
26.             NumeroDeSKUsSeleccionados > NumeroDeElementosMaximosAMostrar,
27.             ListaDeSKUsSeleccionados
28.             & " más otros "
29.             & NumeroDeSKUsSeleccionados - NumeroDeElementosMaximosAMostrar
30.             & " SKUs Seleccionados"
31.             & UNICHAR ( 10 )
32.             & " | Revisar página de segmentación para ver el detalle | ",
33.             ListaDeSKUsSeleccionados
34.         )
35.     )
36.  )
37.  RETURN
38.     Resultado -- Explorar archivo de Power BI asociado para ver la solución.

```

Acerca de las Cartas DAX



Las cartas DAX del equipo de **Excel Free Blog** es un paquete de contenido de documentación y representación para un juego de todas las funciones en lenguaje DAX, compuesta por dos partes:

I. La Carta

Cada función en todo el lenguaje DAX contará con un **personaje representativo**, por ejemplo, la función SUMX será representada por el ser mitológico: el grifo.

II. La Ficha Técnica

La ficha técnica tiene **información de la función** para su manejo, consulta y entendimiento, en ella se documenta y explica: Descripción, sintaxis, parámetros y más. (Cómo la presente)

Más Información

→ <https://bit.ly/3aZiBqu> ←

→ www.CartasDax.Com ←

Última Actualización:

4 de agosto del 2023

Observaciones

- I. La función **CONCATENATEX** admite como máximo 2.1 millones de caracteres concatenados.
- II. La función **CONCATENATEX** es especial en el entorno de las funciones de iteración, ya que es de las pocas que no cumple con los 4 principios de las funciones de iteración, concretamente el principio de *nulidad* y *TRUE/FALSE* no son verdaderos para la función **CONCATENATEX**. Para más información véase la ficha técnica de **MAXX**.

BIBLIOGRAFÍA

Páginas Web:

- 1. DAX GUIDE: <https://dax.guide/concatenatex/>
- 2. MICROSOFT: <https://docs.microsoft.com/en-us/dax/concatenatex-function>
- 3. SQLBI: <https://www.sqlbi.com/articles/using-concatenatex-in-measures/>
- 4. SQLBI: <https://www.sqlbi.com/articles/displaying-nth-element-in-dax/>

Libros:

- Definitive Guide To DAX (2nd Edition) — Marco Russo y Alberto Ferrari [↗](#)
- Practical PowerPivot & DAX Formulas — Art Tennick [↗](#)

Creado por:

Miguel Caballero y Fabian Torres.

Cualquier Retroalimentación:

excelfreebymcs@gmail.com

Funciones Relacionadas:



CONCATENATEX: Dragón de la Iteración Conc.